

# Lidia 80 Digital (2)

## Przystawka do odbiornika homodynowego na pasmo 80 m

**Proste odbiorniki nasłuchowe są wciąż bardzo popularnymi urządzeniami wśród radioamatorów – krótkofalowców. Kosztują niewiele, zajmują mało miejsca i pozwalają przy tym z powodzeniem prowadzić dobre nasłuchy w różnych warunkach. Popularny, łatwy do wykonania odbiornik homodynowy Lidia 80 wg Włodka SP5DDJ został skonstruowany tak, że można go z powodzeniem wykorzystywać w wersji podstawowej, ale daje się też rozbudowywać o kolejne funkcjonalności. Autor tego projektu, zachęcony pozytywnymi doświadczeniami z podstawową wersją odbiornika, postanowił dokonać jego rozbudowy o kolejne funkcje. Apetyt rósł w miarę jedzenia i dlatego ostatnia, prezentowana wersja została wyposażona nie tylko w pomiar częstotliwości pracy, ale także w jej regulację i stabilizację z krokiem minimalnym 10 Hz, graficzny S-metr oraz automatyczną regulację wzmacnienia ARW w zakresie 12 dB (dwie jednostki „S”).**



**Rekomendacje: odbiornik jest przeznaczony dla osób, które chcą zająć się krótkofalarstwem. Prawidłowo zmontowany i zestrojony odbiornik umożliwia nasłuch stacji amatorskich telegraficznych i jednowstęgowych przy użyciu anteny dipolowej lub drutowej (Long-Wire).**

Odpowiednio wyregulowany sygnał m.cz. jest podawany na wejście wzmacniacza wyjściowego z tranzystorem Q9 oraz elementami: C49, R74, R75 i potencjometrem logarytmicznym PR3 (50 kΩ/C). Wzmacniacz ten pracuje w układzie wtórnika emiterowego, a jego zadaniem jest separacja modułu dzielnika regulowanego ARW od dalszej części układu, aby nie obciążała ona dzielnika. Jako PR3 można użyć potencjometru, który służył jako regulator siły sygnału w oryginalnym odbiorniku Lidia 80. Powinien jednak być to podzespół o odpowiednio dobrej jakości, ponieważ przenosi on także składową stałą prądu emitera Q9, co w wypadku potencjometru starego lub zużytego może powodować trzaski przy regulacji głośności. Kondensator C50 separuje dalszą część układu od składowej stałej napięcia na suwaku PR3. Blok ARW i S-metra jest połączony z podstawową (analogową) częścią odbiornika Lidia 80 poprzez port P5, przez który pobierany jest sygnał m.cz. z mieszacza-detektora i przedwzmacniacza w układzie MC3361C (pin 1) a następnie za blokiem ARW

jest podawany na wejście wzmacniacza audio LM386N (pin 2).

Ostatnim, nieomówionym jeszcze blokiem przystawki do odbiornika Lidia 80 są moduły zasilania. Warto tu zwrócić uwagę na fakt, że cały prąd zasilania +12 V, nieprzekraczający wartości 150 mA, jest podawany na port P1 układu i przechodzi przez stabilizator U1 z układem LM7809 i pojemnościami C1...C4. Napięciem stabilizowanym +9 V są zasilane wprost tylko dwa bloki układu: blok ARW i wzmacniacza-detektora do S-metra oraz blok wzmacniacza napięcia sterującego częstotliwością VFO z układem U9 (LM358). Zastosowano stabilizator w obudowie TO-220, dzięki czemu uniknięto konieczności stosowania radiatora. Do wyjścia stabilizatora U1 dołączono równolegle stabilizatory napięcie +5 V (U2 i U3) w obudowach TO-220. Stabilizator U3, z uwagi na znikomy prąd roboczy, jest w obudowie TO-92.

Należy tu wyjaśnić, że połączenie równolegle dwóch stabilizatorów na to samo napięcie nie jest dziełem przypadku czy błędem

projektowym, ale ważnym, celowym zabiegiem konstruktorskim. Na etapie prototypowania układu autor czynił próby zasilania z tego samego stabilizatora U2 także układów U7 i U8 pracujących jako sterowane przez mikrokontroler U4 klucze dla drabinki oporowej R-2R. Niestety, drobne wahania napięcia zasilania +5 V wynikające z impulsowej

REKLAMA

Projekty na na...

# STM32

[www.stm32.eu](http://www.stm32.eu)

ST life.augmented

**KAMAMI**

## Listing 1. Definicje i deklaracje używane w programie

```
'RX Lidia 80 „DIGITAL” - program kontrolujący układ
'Autor: Adam Sobczyk SQ5RWQ 05-10/2015 (e-mail: sq5rwq@gmail.com)
'----- DEFINICJE I DEKLARACJE WSTĘPNE -----
$regfile = „m8def.dat”
$crystal = 16000000
$hwstack = 48 'default use 32 for the hardware stack
$swstack = 20 'default use 10 for the SW stack
$framesize = 40 'default use 40 for the frame space
'PORTY WYJŚCIOWE - STERUJĄCE PRZETWORNIKIEM DAC-16 - DEFINICJE I ALIASY
Config Portd.0 = Output 'DAC SERIAL (SER)
Dac16_serial Alias Portd.0
Config Portd.1 = Output 'DAC SRCLK (SRCLK)
Dac16_srclk Alias Portd.1
Config Portd.2 = Output 'DAC RCLK (RCLK)
Dac16_rclk Alias Portd.2
'PORTY WEJŚCIOWE - DO DETEKCJI KIERUNKU OBRACANIA I STANU PRZYCIŚKI ENKODERA
Config Portb.2 = Input : Portb.2 = 0 'ENCODER F+/R (no internal pull-up)
Config Portb.3 = Input : Portb.3 = 0 'ENCODER F-/L (no internal pull-up)
Config Portb.4 = Input : Portb.4 = 0 'ENCODER DF (no internal pull-up)
'PORTY WYJŚCIOWE - STERUJĄCE ARW (TŁUMIENIE WYŁĄCZONE PO ZDEFINIOWANIU PORTÓW)
Config Portc.0 = Output : Portc.0 = 0 'ARW Q8 DRV
Config Portc.1 = Output : Portc.1 = 0 'ARW Q7 DRV
Config Portc.2 = Output : Portc.2 = 0 'ARW Q6 DRV
Config Portc.3 = Output : Portc.3 = 0 'ARW Q5 DRV
'PORT WEJŚCIOWY DO POMIARU Fvfo - T1 (Timer/Counter1 external counter input)
Config Portd.5 = Input : Portd.5 = 0 'VFO POMIAR F (no internal pull-up)
'PORTY WEJŚCIOWE DO POMIARU POZIOMU SYGNAŁU W.CZ. (S-METR) I NAPIĘCIA ZASILANIA 12V - ADC4, ADC5
Config Portc.4 = Input : Portc.4 = 0 'ARW POMIAR „S” (no internal pull-up)
Config Portc.5 = Input : Portc.5 = 0 'POMIAR U=12V (no internal pull-up)
'KONFIGURACJA I START PRZETWORNIKA ANALOGOWO-CYFROWEGO ADC
Config Adc = Single , Prescaler = Auto , Reference = Internal 'single mode for GETADC() function
Start Adc
'KONFIGURACJA WYŚWIETLACZA ALFANUMERYCZNEGO LCD 16x2 ZNAKI
Config Lcd = 16 * 2
Config Lcdpin = Pin , Db4 = Portd.6 , Db5 = Portd.7 , Db6 = Portb.0 , Db7 = Portb.1 , E = Portd.4 , Rs = Portd.3
'DEFINICJE WŁASNYCH ZNAKÓW DO WYŚWIETLACZA LCD
Deflcdchar 0 , 20 , 20 , 28 , 23 , 21 , 2 , 4 , 7 '„Hz”
Deflcdchar 1 , 16 , 24 , 28 , 25 , 19 , 7 , 3 , 1 '„DF” - SYMBOL KROKU PRZESTRAJANIA
Deflcdchar 2 , 31 , 25 , 23 , 27 , 29 , 19 , 31 , 32 'SPECJALNE „S” DO S-METRA
Deflcdchar 3 , 21 , 21 , 21 , 21 , 21 , 21 , 32 'ZNAK POSTĘPU DO S-METRA DLA 1..3 „S”
Deflcdchar 4 , 27 , 27 , 27 , 27 , 27 , 27 , 32 'ZNAK POSTĘPU DO S-METRA DLA 4..6 „S”
Deflcdchar 5 , 31 , 31 , 31 , 31 , 31 , 31 , 32 'ZNAK POSTĘPU DO S-METRA DLA 7..9 „S”
Deflcdchar 6 , 32 , 31 , 27 , 17 , 27 , 31 , 32 'SPECJALNY ZNAK „+” DO S-METRA DLA >9 „S”

'START EKRANU LCD I EKRAN POWITALNY
Cls
Cursor Off , Noblink
Upperline : Lcd „RX L80 ” ; Chr(34) ; „DIGITAL” ; Chr(34)
Wait 1
Lowerline : Lcd „SQ5RWQ 2015 v1.0”
Wait 3
Cls
'KONFIGURACJA I START LICZNIKA TIMER1 DO POMIARU CZĘSTOTLIWOŚCI VFO
Config Timer1 = Counter , Edge = Rising
On Timer1 Tlover
Enable Ovfl
Stop Timer1
Enable Interrupts
'DEFINICJE ORAZ INICJACJA ZMIENNYCH
Dim Ndac As Word , Dndac As Word 'SŁOWO STERUJĄCE DAC-16 ORAZ KROK JEGO ZMIANY
Ndac = 32768 : Dndac = 1 'NASTAWIENIA POCZĄTKOWE: 50% ZAKRESU DAC-16, MINIMALNY KROK ZMIANY
Dim F As Long 'WYNIK POMIARU CZĘSTOTLIWOŚCI Fvfo
Dim Fc As Long , Dfc As Word 'NASTAWIANA CZĘSTOTLIWOŚĆ ORAZ KROK JEJ STROJENIA
Fc = 3710000 : Dfc = 10000 'NASTAWIENIA POCZĄTKOWE: „POLSKIE” PASMO KF SSB, KROK STROJENIA 10kHz (ZMIENI SIĘ
NA 1kHz PO STARCIE PROGRAMU)
Dim Df As Long 'BŁĄD CZĘSTOTLIWOŚCI VFO
Dim Uz As Word , Uvolt As Single 'Uz - POMIAR NAP. ZAS. 12V, Uvolt - PRZELICZONA WARTOŚĆ DO WYŚWIETLENIA NA LCD
Dim S As Word , Sdb As Single , Sout As Byte 'S - POMIAR SIŁY SYGNAŁU W.CZ., Sdb - S W SKALI LOG. [dB], Sout - WARTOŚĆ
W JEDNOSTKACH „S”
Dim Sarw As Single , Arw As Byte 'Sarw - ZM. POMOCCNICZA DO WYZNACZENIA SŁOWA „ARW”, STERUJĄCEGO TYM MECHANIZMEM
Dim Txt As String * 16 'CIĄG TEKSTOWY DO FORMATOWANIA WYŚWIETLANIA NASTAWIANEJ CZĘSTOTLIWOŚCI VFO Fc
(Z MIEJSCEM NA SPACJĘ NA KOŃCU)
Dim Delayword As Word 'ZMIENNA DO BRAMKOWANIA CZASU POMIARU Fvfo PRZEZ TIMER1
```

pracy mikrokontrolera U4 i wyświetlacza LCD (U5) skutecznie przenosiły się na zasilanie układów scalonych U7 i U8, które stanowi jednocześnie napięcie referencyjne Uref dla przetwornika C/A. Na skutek tego zjawiska impulsy prądu zasilania U4 i U5 przenoszące się na napięcie zasilania U7 i U8 przechodziły też na wyjście przetwornika C/A i silnie modulowały częstotliwość VFO (było to wyraźnie słyszalne w głośniku odbiornika). Próby całkowitego wyeliminowania opisanych zakłóceń za pomocą włączonego w tor zasilania układów U7 i U8 niewielkiego rezystora szeregowego, zablokowanego do masy po stronie odbioru prądu nawet bardzo dużymi pojemnościami (rzędu kilku tysięcy mikrofaradów), w ogóle nie pomagały zmierzyć się z opisanym problemem. Skutecznym

rozwiązaniem okazało się dopiero zastosowanie osobnego stabilizatora napięcia (U3), przeznaczonego wyłącznie do zasilania układów U7 i U8.

W trosce o całkowitą eliminację przenikania między poszczególnymi blokami układu przez tory ich zasilania zakłóceń impulsowych i w.cz. zastosowano dodatkowe elementy odprężające R-C. Są nimi rezystory R1, R2, R52 i R56 o wartościach dobranych tak, aby spadek napięcia na nich nie zaburzał pracy zasilanych bloków. Rezystory po stronie odbioru prądu zostały zablokowane do masy dla składowej zmiennej za pomocą odpowiedniego zestawu pojemności (kondensator elektrolityczny o większej wartości i ceramiczny lub foliowy o wartości mniejszej, skuteczny dla w.cz.).

Schemat ideowy połączenia w jedną całość odbiornika Lidia 80 z przystawką cyfrową będzie omówiony szczegółowo w części poświęconej montażowi i uruchomieniu urządzenia. W tym miejscu warto jeszcze podkreślić to, że płyta cyfrowej przystawki jest połączona z elementami zewnętrznymi za pomocą portów:

- P1 (zasilanie +12 V, podawane na złącze ARK-2).
- P2 (5-pinowe złącze „goldpin” do enkodera z przyciskiem).
- P3 (2-pinowe złącze „goldpin” do pobierania sygnału do pomiaru częstotliwości sygnału z generatora VFO).
- P4 (2-pinowe złącze „goldpin” do sterowania częstotliwością pracy generatora VFO).

- P5 (3-pinowe złącze „goldpin” przeznaczone do włączenia bloków ARW i S-metra w tor audio oryginalnego odbiornika Lidia 80).
- Wyprowadzenia do potencjometru PR3 (także 3-pinowe złącze „goldpin”), przeznaczonego do regulacji głośności odbioru.

## Oprogramowanie mikrokontrolera sterującego

Program sterujący napisano w języku Bascom AVR. **Listing 1** zawiera definicje i deklaracje wstępne, niezbędne do dalszej pracy programu. Pierwsze linijki służą do konfiguracji pracy mikrokontrolera U4 (ATmega-8A-PU). Zdefiniowano typ i częstotliwość taktowania MCU zewnętrznym rezonatorem kwarcowym (16 MHz), a także rozmiary stosów i ramki programu. Następnie zdefiniowane zostały jako wyjściowe porty PD.0...PD.2, sterujące pracą przetwornika C/A z układami U7 i U8 za pomocą transmisji szeregowej. W kolejnym kroku określono jako wejściowe porty PB.2...PB.4, których zadaniem jest odczytywanie z bloku z układem scalonym U6 informacji o stanie przycisku oraz o kierunku obracania enkodera, ustalającego częstotliwość pracy generatora VFO. Kolejne definicje dotyczą portów wyjściowych PC.0...PC.3, sterujących kluczami elektronicznymi z tranzystorami Q5...Q8 w bloku automatycznej regulacji wzmocnienia ARW. Z kolei port PD.5 został określony jako wejście do pomiaru częstotliwości Fvfo z wykorzystaniem licznika TIMER1 („Timer/Counter1 external counter input”). Ostatnie zdefiniowane dwa porty to PC.4 i PC.5, które jako komutowane wejścia ADC4 i ADC5 wewnętrznego przetwornika A/C ATmega służą w tym układzie do pomiaru poziomu sygnału w.c. (na potrzeby S-metra i regulacji ARW) oraz do pomiaru napięcia zasilania układu  $Uz=12V$ .

W kolejnych krokach następuje skonfigurowanie do pracy oraz uruchomienie (początek gotowości do użycia) przetwornika A/C. Z założenia ma on pracować w trybie pojedynczego wyzwania pomiaru („on-demand”) – z automatycznym preskalarem zegara oraz z wykorzystaniem wewnętrznego źródła napięcia odniesienia  $U_{ref}=2,56V$ . W następnych linijkach zdefiniowano sposób dołączenia wyświetlacza alfanumerycznego LCD oraz znaki specjalne dla tego ekranu. Zdefiniowano jednoznakowy symbol jednostki „Hz” (symbol o numerze 0), symbol kroku strojenia częstotliwości (1), symbol jednostki „S” dla bargrafu S-metra (2), a także cztery symbole (3...6), potrzebne do uformowania bargrafu na wyświetlaczu LCD (trzy rodzaje pionowych „beleczek” oraz specjalny symbol „+”).

Po wyświetleniu ekranu powitalnego w programie następuje konfiguracja licznika Timer1 (wykorzystywanego do pomiaru

częstotliwości VFO). Aktywowane zostaje także przerwanie wynikające z przepelnienia tego licznika – wraz z odpowiednią procedurą jego obsługi T1over, omówioną bliżej przy głównej procedurze pomiaru Fvfo.

Ostatni blok listingu 1 to definicje wszystkich używanych w programie zmiennych. W skrócie ich przeznaczenia są następujące:

- Ndac, Dndac – zmienne bezpośrednio wspierające proces sterowania przetwornikiem C/A.
- F, Fc, Dfc, Df, Txt, Delayword – zmienne wykorzystywane w zadaniach związanych z pomiarem, regulacją oraz wyświetlaniem częstotliwości pracy VFO.
- Uz, Uvolt – zmienne zaangażowane w pomiar i prezentację na wyświetlaczu LCD napięcia zasilania układu Uz.
- S, Sdb, Sout – zmienne pracujące w procesach pomiaru, przetwarzania i wizualizacji poziomu odbieranego sygnału w.c. na wejściu odbiornika.
- Sarw, Arw – zmienne bezpośrednio użyte w mechanizmie sterowania blokiem ARW.

Na **listingu 2** pokazano blok programu sterującego całym urządzeniem. Po zainicjowaniu przetwornika C/A za pomocą podprogramu *Dac16\_set* (pierwsze nastawienie wartości wyjściowej – w połowie dostępnego zakresu napięć) oraz wyświetleniu informacji o nastawianej częstotliwości VFO i kroku jej regulacji (podprogramy: *Shownewfc* i *Changedfc*) następuje przejście do powtarzanej nieskończenie głównej pętli programu *Do-Loop*. W pętli tej cyklicznie powtarzane są zadania programu.

Pierwsze z nich to ciąg instrukcji warunkowych *If-Then-Elseif* wykrywających, czy występuje obracanie pokrętką enkodera. Jeśli tak i jeśli nastawiona częstotliwość nie znajduje się akurat na którymś z krańców pasma 3,5...3,8 MHz, to nastąpi zmiana wartości  $F_c$  o jeden krok  $D_{fc}$ . Dalej następuje pomiar (podprogram *Fmeasure*) i ewentualna korekta wartości  $F_{vfo}$  za pomocą podprogramu *Dac16\_set*. W celu dokonania korekty częstotliwości  $F_{vfo}$  najpierw jest wyliczana różnica pomiędzy zmierzoną częstotliwością  $F$  a częstotliwością zadaną  $F_c$ . Następnie, na podstawie jej wartości bezwzględnej  $D_f$ , obliczona zostaje wartość poprawki słowa *Ndac*, sterującego przetwornikiem C/A równa  $D_{ndac} = D_f \setminus 10$  (dzielenie całkowitoliczbowe). Ostatecznie, w zależności od znaku rozbieżności częstotliwości, poprawka *Dndac* jest dodawana lub odejmowana od słowa sterującego *Ndac*, ładowanego następnie do przetwornika C/A podprogramem *Dac16\_set*.

Na wyjaśnienie zasługuje sposób wyliczenia poprawki słowa sterującego  $D_{ndac} = D_f \setminus 10$ . Zależność  $N_{dac} = f(F_{vfo})$  przybliżono funkcją liniową  $y = 0,1433 \cdot x - 487106$ .

## DODATKOWE MATERIAŁY NA FTP:

<ftp://ep.com.pl>

USER: 11875, PASS: 6hhcxtt

W ofercie AVT\*

## AVT-5527 A, UK

Projekty pokrewne na FTP:

(wymienione artykuły są w całości dostępne na FTP)

AVT-5527	Lidia 80 Digital (EP 1/2016)
---	Odbiornik nasłuchowy na pasmo 80 m (EP 6/2015)
AVT-2970	Odbiornik SDR na pasmo 2 m (EdW 2/2011)
AVT-2960	Minitransceiver SP5AHT (80 m/SSB) (EdW 11/2010)
AVT-2934	Odbiornik na pasmo 80 m (EdW 2/2010)
AVT-2925	Odbiornik nasłuchowy „Cypisek” (EdW 12/2009)
AVT-2902	Wzmacniacz mocy na pasmo 80 m (EdW 6/2009)
AVT-2891	Prosty odbiornik nasłuchowy na pasmo 80 m (EdW 2/2009)
AVT-5151	Minitransceiver „JędreK” (EP 10/2008)
AVT-5127	Minitransceiver na pasmo 3,7 MHz TRX2008 (EP 3-4/2008)
AVT-967	Minitransceiver Junior (EP 2/2007)
AVT-962	Odbiornik nasłuchowy SSB/CW 80 m (EP 1/2007)

\* Uwaga: Zestawy AVT mogą występować w następujących wersjach:  
 AVT xxxx UK to zaprogramowany układ. Tylko i wyłącznie. Bez elementów dodatkowych.  
 AVT xxxx A płytka drukowana PCB (lub płytki drukowane, jeśli w opisie wyraźnie zaznaczono), bez elementów dodatkowych.  
 AVT xxxx A- płytka drukowana i zaprogramowany układ (czyli połączenie wersji A i wersji UK) bez elementów dodatkowych.  
 AVT xxxx B płytka drukowana (lub płytki) oraz komplet elementów wymienionych w załączniku pdf  
 AVT xxxx C to nic innego jak zmontowany zestaw B, czyli elementy wmontowane w PCB. Należy mieć na uwadze, że o ile nie zaznaczono wyraźnie w opisie, zestaw ten nie ma obudowy ani elementów dodatkowych, które nie zostały wymienione w załączniku pdf  
 AVT xxxx CD oprogramowanie (niezwykle spotykana wersja, lecz jeśli występuje, to niezbędne oprogramowanie można ściągnąć, klikając w link umieszczony w opisie kitu)  
 Nie każdy zestaw AVT występuje we wszystkich wersjach! Każda wersja ma załączony ten sam plik pdf! Podczas składania zamówienia upewnij się, którą wersję zamawiasz! (UK, A, A-, B lub C). <http://sklep.avt.pl>

Odwrotność współczynnika proporcjonalności (przy zmiennej  $x$ ) wynosi  $1/0,1433=6,978$ , czyli około 7. Oznacza to, że zgodnie z podanym przybliżeniem zmiana częstotliwości VFO o 7 Hz wymaga korekty słowa sterującego *Ndac* w przybliżeniu o jedność (zmiany ułamkowe z oczywistych powodów nie są możliwe). Jakkolwiek, z uwagi na niedoskonałość tej aproksymacji liniowej oraz chcąc zachować możliwie jak najmniejsze kroki korekt  $F_{vfo}$  ( $D_{ndac}=1$  dla  $D_f=10$  Hz,  $D_{ndac}=2$  dla  $D_f=20$  Hz,  $D_{ndac}=3$  dla  $D_f=30$  Hz itd.), ostatecznie postanowiono zwiększyć wykorzystany odwrócony współczynnik proporcjonalności z 7 do 10. Takie podejście gwarantuje wysoką stabilność i precyzję regulacji  $F_{vfo}$ , jednocześnie nie

REKLAMA

Projekty na...

# STM32



[www.stm32.eu](http://www.stm32.eu)

**ST** **KAMAMI**

life.augmented

powodując zauważalnego jej spowolnienia w trakcie normalnej pracy.

Kolejny krótki fragment kodu to instrukcja warunkowa *If-Then*, która w przypadku wykrycia naciśnięcia przycisku enkodera obrotowego wywoła podprogram *Changedfc* zmieniający 10-krotnie krok strojenia częstotliwości VFO *Dfc* w zakresie od 10 Hz do 10 kHz.

W następnym kroku realizowany jest pomiar wolnozmiennego napięcia, niezbędnego doysterowania modułów ARW i S-metra. Jest ono pobierane z suwaka potencjometru montażowego PR2, który służy do kalibracji (skalowania) wymienionych modułów. Zmierzona w przetworniku A/C wartość (w programie: zmienna *S*) jest zwiększana o 1, by uniknąć logarytmowania

z zera. Następnie, po rzutowaniu *S* (słowo 16-bitowe) na zmiennoprzecinkową zmienną *Sdb* następuje przeliczenie jej wartości na skalę logarymiczną (decybele [dB]) i – finalnie – na skalę wartości „prawdziwych” jednostek „*S*” w zmiennej *Sout* (jednostka „*S*” to około 6 dB). Przyjmując opisany sposób wyliczenia zmiennej *Sout*, otrzymujemy mapowanie pierwotnej zmiennej *S* (w zakresie od 0 do 1023 na wyjściu 10-bitowego przetwornika A/C) na zmienną *Sout* (w zakresie od 0 do 10 jednostek „*S*”), które po zaokrągleniu do najbliższej wartości całkowitej wygląda tak, jak na **rysunku 4** (dodatkowa linia przerywana, to logarymiczna linia trendu, obrazująca średnie wartości *Sout* przed zaokrągleniem).

Kolejny blok głównego programu przeznaczony jest do sterowania mechanizmem ARW w zakresie 12 dB (od 7 do 9 „*S*”). Mając na uwadze sposób wyliczenia zmiennej *Sout*, za dolny próg zadziałania ARW należy przyjmując wartość zmiennej *S*=128. Po przesunięciu „w dół” zmierzonego poziomu sygnału odbieranego w dB (zmienna *Sdb*) o około 42 dB, czyli 7 jednostek „*S*”, oraz po przełożeniu wartości otrzymanej różnicy na aproksymowaną funkcją  $y=0,0604*x^2+0,4926*x=x*(0,0604*x+0,4926)$  „odwróconą” charakterystykę sterowania ARW, otrzymujemy wstępną, zmiennoprzecinkową wartość słowa sterującego *Sarw*. Wartość zmiennej *Sarw*, po ograniczeniu do wartości akceptowalnych dla 4-bitowego sterowania blokiem ARW (0...15) oraz zaokrągleniu

## Listing 2. Główny blok programu

```
Gosub Dac16_set 'INICJACJA PRZETWORNIKA DAC-16, PRESET VFO
Gosub Shownewfc : Gosub Changedfc 'INICJACJA WYŚWIETLANIA NASTAWY I KROKU ZMIANY CZĘSTOTLIWOŚCI VFO

'GŁÓWNA PĘTLA PROGRAMU
Do
  'ZMIANA USTAWIENIA Fc
  If Pinb.3 = 0 And Fc > 3500000 Then 'OBRÓT ENKODERA W LEWO, WARTOŚĆ Fc JEST W DOZWOLONYM ZAKRESIE
    Fc = Fc - Dfc
    Gosub Shownewfc
  ElseIf Pinb.2 = 0 And Fc < 3800000 Then 'OBRÓT ENKODERA W PRAWO, WARTOŚĆ Fc JEST W DOZWOLONYM ZAKRESIE
    Fc = Fc + Dfc
    Gosub Shownewfc
  End If

  'POMIAR I EWENTUALNA KOREKTA WARTOŚCI Fvfo
  Gosub Fmeasure 'POMIAR Fvfo
  Df = F - Fc : Df = Abs(df) 'WYZNACZENIE WARTOŚCI BEZWZGLEDNEJ ODCHYLENIA CZĘSTOTLIWOŚCI
  Dndac = Df \ 10 'WYZNACZENIE KROKU REGULACJI
  If F > Fc Then
    Ndac = Ndac - Dndac
    Gosub Dac16_set
  ElseIf F < Fc Then
    Ndac = Ndac + Dndac
    Gosub Dac16_set
  End If

  'EWENTUALNA ZMIANA KROKU REGULACJI CZĘSTOTLIWOŚCI VFO dFC
  If Pinb.4 = 0 Then 'NACIŚNIĘTO PRZYCIŚK ENKODERA
    Gosub Changedfc
  End If

  'POMIAR NAPIĘCIA DLA ARW I S-METRA - DO DALSZEJ KALIBRACJI POTENCJOMETREM PR2
  S = Getadc(4) : Incr S 'KOREKTA +1, BY UNIKNAĆ DALEJ LOGARYTMOWANIA Z ZERA
  Sdb = S 'RZUTOWANIE ZMIENNEJ S Z TYPU WORD NA TYP SINGLE - DO OBLICZEŃ ZMIENNOPRZECINKOWYCH
  Sdb = 20 * Log10(sdb) 'PRZELICZENIE NA SKALĘ LOGARYMICZNĄ [dB]
  Sout = Sdb / 6.0206 'ORAZ NA SKALĘ WARTOŚCI „S” DLA S-METRA (1S = ok. 6 dB)

  'STEROWANIE ARW W ZAKRESIE 12 dB (S7..S9)
  If S >= 128 Then 'PRZYJMujemy POZIOM 7 „S” (S=126) ZA PRÓG ZADZIAŁANIA ARW
    'ŚREDNI POZIOM 7 „S” (S=128 po dodaniu 1 do wartości zmierzonej w ADC), TO ok. 42 dB PONAD POZIOM S0
    'W ARW JEDNAK INTERESUJE NAS DELTA PONAD POZIOM 7 „S”)
    Sdb = Sdb - 42.144
    'APROKSYMACJA CHARAKTERYSTYKI STEROWANIA ARW WG FUNKCJI: y = 0.0604*x^2 + 0.4926*x = X*(0.0604*X + 0.4926)
    Sarw = 0.0604 * Sdb
    Sarw = Sarw + 0.4926
    Sarw = Sdb * Sarw
    'WYLICZENIE CAŁKOWITEGO 4-BITOWEGO SŁOWA STERUJĄCEGO ARW
    If Sarw > 15 Then
      Arw = 15
    Else
      Arw = Round(sarw)
    End If
  Else
    Arw = 0
  End If
  Portc = Arw 'WYSTEROWANIE BLOKU ARW

  'PREZENTACJA WARTOŚCI „S” NA LCD ZA POMOCĄ BARGRAFU Z RÓŻNYMI SYMBOLAMI POSTĘPU
  Lowerline
  Select Case Sout
    Case Is <= 0 : Lcd Spc(9) ; Chr(2)
    Case 1 To 3 : Lcd String(sout , 3) ; Spc(9 - Sout) ; Chr(2)
    Case 4 To 6 : Lcd String(3 , 3) ; String(sout - 3 , 4) ; Spc(9 - Sout) ; Chr(2)
    Case 7 To 8 : Lcd String(3 , 3) ; String(3 , 4) ; String(sout - 6 , 5) ; Spc(9 - Sout) ; Chr(2)
    Case 9 : Lcd String(3 , 3) ; String(3 , 4) ; String(3 , 5) ; Chr(2)
    Case Is > 9 : Lcd String(3 , 3) ; String(3 , 4) ; String(3 , 5) ; Chr(6)
    Case Else : Lcd String(sout , 45)
  End Select

  'POMIAR I PREZENTACJA NAPIĘCIA ZASILANIA Uz=12V
  Uz = Getadc(5)
  Uvolt = 0.0166026 * Uz 'SKALOWANIE Uz WZGLĘDEM Uref I WSP. PODZIAŁU DZIELNIKA R3-R4
  Locate 2 , 11 : Lcd „ ” ; Fusing(uvolt , „#.##”) ; „V ”
Loop
```


do najbliższych wartości całkowitych i rzutowaniu na zmienną typu *byte* (*Arw*), wpisujemy do czterech najmłodszych bitów portu C mikrokontrolera. W praktyce, dla przyspieszenia i uproszczenia działania programu, zrealizowano wpis całego bajtu do wszystkich dostępnych rejestrów portu C, jednak na to działanie zareagują tylko wskazane cztery najmłodsze bity, skonfigurowane jako wyjściowe.

Ostatnie dwa bloki głównej pętli programu kontrolującego układ to wizualizacja siły sygnału odbieranego oraz pomiar i prezentacja wartości napięcia zasilania *Uz*. Rezultaty obu tych działań zobaczymy w dolnej linijce wyświetlacza LCD.

Siła sygnału, reprezentowana przez zmienną *Sout*, wyrażoną w jednostkach „S”, jest wyświetlana za pomocą bargrafu, czyli ruchomego paska, którego długość zmienia się wraz z wartością *Sout*. Aby uatrakcyjnić wygląd i zwiększyć czytelność tej wizualizacji, zastosowano zaprojektowane we własnym zakresie znaki symbolizujące siłę sygnału. Za pomocą bloku instrukcji *Select-Case*, w zależności od wartości zmiennej *Sout* wyświetlana jest różna długość linijki poziomej, przy czym zastosowano odrębne symbole dla wartości „S” od 1 do 3, od 4 do 6 oraz od 7 do 9. Dodatkowo, po przekroczeniu wartości siły sygnału na poziomie 9 „S” na końcu linijki zamiast symbolu jednostki wyświetlany jest specyficzny symbol „+” (w negatywie), sygnalizujący fakt, że mamy do czynienia z silnym sygnałem – z tzw. „plusami” (9 „S” + N dB).

Na końcu głównej pętli *Do-Loop* następuje pomiar napięcia zasilania *Uz* (za dzielnikiem rezystancyjnym R3-R4). Pomiaru dokonuje przetwornik analogowo-cyfrowy w ATmedze poprzez multipleksowane wejście ADC5 (PORTC.5 mikrokontrolera). Zamiast uśrednienia kilku pomiarów (co zwykle pomaga zwiększyć ich dokładność i stabilność) zastosowano rozwiązanie układowe, którym jest dodanie pojemności C12 (10  $\mu$ F), silnie filtrującej dolnoprzepustowo mierzony sygnał. Ostatecznie wynik pomiaru jest skalowany przez współczynnik proporcjonalności  $A=(2,56 \text{ V}/1024)/[3,9 \text{ k}\Omega/(22 \text{ k}\Omega+3,9 \text{ k}\Omega)]=0,0166026 \text{ V/bit}$  i – po ograniczeniu do jednej cyfry po przecinku z użyciem wygodnej do tego celu funkcji *Fusing* – wyświetlany na wyświetlaczu LCD.

**Listing 3** obejmuje wszystkie podprogramy, wykorzystywane przez program główny. Ich wywołanie następuje każdorazowo za pomocą instrukcji *Gosub* (skok z powrotem do miejsca wywołania). Dzięki takiemu podejściu całkowicie uniknięto problemu przekazywania danych wejściowych do podprogramów oraz zwracania rezultatów ich działania – wszystkie zmienne w całym programie mają charakter globalny i przez to są dostępne na jednakowych zasadach w każdej części omawianego kodu. Podprogramy zawsze zaczynają się po prostu nazwą etykiety, czyli identyfikatora miejsca w programie (ang. *label*), a kończą słowem



**Rysunek 4. Mapowanie zmiennej S (w zakresie od 0 do 1023 na wyjściu 10-bitowego przetwornika A/C) na zmienną Sout**

kluczowym *Return*, nakazującym powrót do miejsca swojego wywołania.

Pierwszy podprogram to *Dac16\_set*, którego zadaniem jest ustawianie właściwej (aktualnej) wartości 16-bitowego słowa sterującego *Ndac* na wyjściu przetwornika DAC. Zadanie to na drodze transmisji szeregowej realizuje polecenie *Shiftout*, którego parametrami są kolejno: nazwa portu, na którego wyjście należy szeregowo przekazać dane binarne, nazwa portu, gdzie należy podać zegar (taktowanie) transmisji szeregowej, nazwa zmiennej (słowa binarnego) do przesłania oraz zastosowany tryb transmisji. W podprogramie skonfigurowano tryb transmisji numer 1. Po załadowaniu całego 16-bitowego słowa do rejestrów układów U7 i U8 na około 5 mikrosekund ich wejścia RCLK są podnoszone do logicznego poziomu wysokiego, co ma na celu przeładowanie świeżo załadowanych 16 bitów danych z wewnętrznego, szeregowego rejestru przesuwanego do rejestru wyjściowego. Po wykonaniu tej operacji słowo *Ndac* przenosi się na 16 wyjść logicznych układów U7 i U8 i na drabinkę rezystorową R-2R, a finalnie na port P4 sterujący częstotliwością pracy VFO.

Kolejny podprogram nazwany *Fmeasure* służy do pomiaru częstotliwości *Fvfo* z rozdzielczością 10 Hz. W pierwszej kolejności jest zerowana zmienna *F* wykorzystywana do przechowywania obliczeń pośrednich oraz do przekazania poza tę procedurę końcowego rezultatu pomiaru. Następnie jest zerowany 16-bitowy licznik *Timer1*, który po inicjacji programu pozostaje w stanie zatrzymania. Następnie *Timer1* jest uruchamiany i zaczyna zliczać impulsy z wyjścia VFO uformowane do poziomów TTL. W tym momencie rozpoczyna się działanie bloku opóźniającego, którego zadaniem jest wstrzymanie na około 100 ms pracy całego mikrokontrolera – poza wspomnianym licznikiem impulsów i przerwaniem, które z nim współpracuje. Ten „czas martwy” mikrokontrolera stanowi niejako

wzorec czasu (bramkę) dla pomiaru z dokładnością do 10 Hz. Opóźnienie zrealizowano za pomocą pustej pętli *For-Next*, wewnątrz której nie są wykonywane żadne instrukcje. Doświadczalnie sprawdzono, że jedno wykonanie takiej pustej pętli, taktowanej 16-bitową zmienną całkowitoliczbową typu *Word*, zabiera dokładnie 33 cykle zegara procesora. Przy zastosowaniu taktowania zegara kwarcem o częstotliwości 16 MHz, w celu uzyskania opóźnienia 100 ms, jest konieczne wykonanie 48456 takich „pustych pętli”, a brakujący czas uzupełniono eksperymentalnie 27 jednocyfrowymi rozkazami mikroprocesora *NOP*. Warto podkreślić, że czysto deterministyczne (bez podejścia eksperymentalnego) liczby przejść „pustej” pętli *For-Next* oraz liczby uzupełniających rozkazu *NOP* nie jest możliwe bez dokładnej wiedzy o tym, ile czasu (cykli zegara) upłynie pomiędzy: faktycznym startem licznika *TIMER1* a zakończeniem komendy wykonującej to zadanie, inicjacją pętli *For-Next* a realną chwilą rozpoczęcia iteracji oraz rozpoczęciem wykonywania instrukcji zatrzymującej *TIMER1* a rzeczywistym jego zatrzymaniem. W trakcie zliczania impulsów każde przepełnienie 16-bitowego licznika wywołuje stosowne przerwanie *T1 overflow interrupt handling*, które obsługuje

REKLAMA

Projekty na...Texas  
**STM32**  
 www.stm32.eu  
 life.augmented  
 KAMAMI

kolejny podprogram, umieszczony za etykietą T1over. Obsługa tego przerwania polega jedynie na zwiększeniu o jeden wyjściowej zmiennej *F*. Po zatrzymaniu licznika wartość zmiennej *F* jest mnożona przez  $2^{16}$  dla uwzględnienia faktu, że każde przepełnienie tego licznika oznacza jednokrotne zliczenie takiej właśnie liczby impulsów. W tym konkretnym wypadku operację mnożenia dało się wygodnie zastąpić przesunięciem bitowym o 16 pozycji w lewo.

Na koniec podprogramu pomiaru częstotliwości do wartości *F* doliczane są impulsy zliczone w liczniku tuż przed zamknięciem wzorcowego okna czasu, a tak uzyskany wynik jest mnożony przez 10 (skalowanie niezbędne z uwagi na fakt, że zastosowany wzorzec czasu trwa 1/10 sekundy). Należy w tym miejscu wyjaśnić, że do odmierzania czasu zliczania impulsów („bramki” licznika) nie użyto innego licznika-timera po to, aby uniknąć konfliktu przerwania od obu liczników, który mógłby skutkować błędami w pomiarze *Fvfo*.

Podprogram *Changedfc* obsługuje wciśnięcie przycisku zmiany kroku regulacji częstotliwości w enkoderze. Krok strojenia *Dfc* jest zmieniany w ramach pętli nieskończonej  $\dots \rightarrow 10 \text{ kHz} \rightarrow 1 \text{ kHz} \rightarrow 100 \text{ Hz} \rightarrow 10 \text{ Hz} \rightarrow 10 \text{ kHz} \dots$  Funkcjonuje to poprzez podział wartości *Dfc* przez 10, a po napotkaniu minimalnej dozwolonej wartości (10 Hz) następuje powrót do najwyższej możliwej wartości 10 kHz. Aktualizacja wartości kroku strojenia VFO na wyświetlaczu LCD następuje z wykorzystaniem instrukcji *Select-Case* i specjalnie w tym celu zdefiniowanych symboli (patrz: listing 1) kroku strojenia oraz jednostki Hz. Na koniec opisywanej procedury wprowadzono opóźnienie 200 ms po to, by nieco zbyt długie naciśnięcie przycisku nie powodowało szybkiej cyklicznej zmiany kroku przestrajania.

Ostatni zdefiniowany podprogram *Shownewfc* służy do obsługi zmiany nastawienia częstotliwości VFO. W pierwszej kolejności realizowane jest sprawdzenie tego, czy

nie przekroczono dozwolonych granic częstotliwości VFO (3,5...3,8 MHz). Jeśli tak, to wartość częstotliwości *Fc* jest przywracana na dozwolone granice, co chroni także główny program przed próbą nastawienia w VFO niedozwolonej częstotliwości. Potem następuje uaktualnienie wartości częstotliwości *Fc* na wyświetlaczu LCD. W celu zwiększenia estetyki i czytelności tego odczytu zrezygnowano z dostępnych w kompilatorze gotowych funkcji formatujących na rzecz własnej prostej procedury. Dzięki tej prostej funkcji formatującej pozycje cyfr, oznaczające MHz, kHz i pojedyncze Hz, są oddzielone kropkami a liczba cyfr dla pojedynczych herców została ograniczona do dwóch (tylko setki i dziesiątki), co odpowiada uzyskanej w urządzeniu dokładności pomiaru i regulacji częstotliwości VFO (10 Hz).

**Adam Sobczyk SQ5RWQ**  
[sq5rwq@gmail.com](mailto:sq5rwq@gmail.com)  
<http://sq5rwq.pl>

### Listing 3. Pozostałe funkcje pomocnicze

```
'PRZETWORNIK DAC-16 - USTAWIENIE WARTOŚCI WYJŚCIOWEJ (TRYB: „MSB shifted out first when clock goes high”)
Dacl6_set:
    Shiftout Dacl6_serial , Dacl6_srclk , Ndac , 1
    Set Dacl6_rclk : Waitus 5 : Reset Dacl6_rclk    ': Waitus 5
Return

'POMIAR CZĘSTOTLIWOŚCI Fvfo Z ROZDZIELCZOŚCIĄ 10 Hz
Fmeasure:
    F = 0          'KASOWANIE WYNIKU ZLICZANIA Fvfo
    Timer1 = 0     'KASOWANIE LICZNIKA IMPULSÓW
    Start Timer1   'START LICZNIKA IMPULSÓW

    'BLOK OPÓŹNIAJĄCY (100ms) - BRAMKUJĄCY OKNO CZASOWE LICZNIKA IMPULSÓW
    'ZGRUBNE OPÓŹNIENIE W PUSTEJ PĘTLI FOR-NEXT (JEDNO WYKONANIE, TO OK. 33 CYKLE MCU Z ZEGAREM 16 MHz)
    For Delayword = 1 To 48456
    Next Delayword
    'PRECYZYJNE UZUPEŁNIENIE OKNA CZASOWEGO 27 „PUSTYMI” 1-CYKLOWYMI INSTRUKCJAMI NOP (NO OPERATION)
    NOP : NOP : NOP : NOP : NOP : NOP : NOP : NOP : NOP
    NOP : NOP : NOP : NOP : NOP : NOP : NOP : NOP : NOP
    NOP : NOP : NOP : NOP : NOP : NOP : NOP : NOP : NOP

    'PODSUMOWANIE CYKLU ZLICZANIA
    Stop Timer1   'ZATRZYMANIE ZLICZANIA
    Shift F , Left , 16 'SZYBKI ODPowiednik pomnożenia wyniku zliczania impulsów przez 65536
    F = F + Timer1 'UZUPEŁNIENIE WYNIKU ZLICZANIA O TO, CO POZOSTAŁO W LICZNIKU TIMER1
    F = F * 10     'PRZELICZENIE WYNIKU NA [Hz] - Z UWAGI NA ZASTOSOWANE CZASOWE OKNO BRAMKUJĄCE T=0.1sek.
Return

'OBŚLUGA PRZERWANIA OD PRZEPEŁNIENIA LICZNIKA TIMER1 („T1 overflow interrupt handling”)
T1over:
    Incr F
Return

'OBŚLUGA PRZYCIŚNIĘCIA PRZYCIŚNIĘCIA ZMIANY KROKU REGULACJI CZĘSTOTLIWOŚCI W ENKODERZE
Changedfc:
    'Z ZASADY ZMNIEJSZAMY KROK 10X TAK DŁUGO, JAK TO TYLKO MOŻLIWE (ZAPĘTLENIE: ..->10 kHz->1 kHz->100 Hz->10 Hz->10 kHz..)
    If Dfc > 10 Then
        Dfc = Dfc \ 10
    Else
        Dfc = 10000
    End If
    'AKTUALIZACJA WARTOŚCI KROKU CZĘSTOTLIWOŚCI NA WYŚWIETLACZU LCD
    Locate 1 , 12 : Lcd Chr(1)
    Select Case Dfc
        Case 10 : Lcd „10” ; Chr(0) ; „ ”
        Case 100 : Lcd „100” ; Chr(0) ; „ ”
        Case 1000 : Lcd „1k” ; Chr(0) ; „ ”
        Case 10000 : Lcd „10k” ; Chr(0) ; „ ”
    End Select
    'OPÓŹNIENIE NIEZBEDNE, BY ZMIANA KROKU CZĘSTOTLIWOŚCI NIE NASTĘPOWAŁA KILKA RAZY ZA JEDNYM PRZYCIŚNIĘCIEM
    Waitms 200
Return

'OBŚLUGA ZMIANY NASTAWY CZĘSTOTLIWOŚCI VFO
Shownewfc:
    'OBŚLUGA PRZEKROCZENIA DOZWOLONYCH GRANIC DLA CZĘSTOTLIWOŚCI VFO
    If Fc < 3500000 Then
        Fc = 3500000
    Elseif Fc > 3800000 Then
        Fc = 3800000
    End If
    'AKTUALIZACJA WARTOŚCI CZĘSTOTLIWOŚCI NA WYŚWIETLACZU LCD
    Uppeline
    Txt = Str(fc)
    Lcd Left(txt , 1) ; „.” ; Mid(txt , 2 , 3) ; „.” ; Mid(txt , 5 , 2) ; „M” ; Chr(0) ; „ ”
Return
```